NAG Fortran Library Routine Document

G13DNF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G13DNF calculates the sample partial lag correlation matrices of a multivariate time series. A set of χ^2-statistics and their sigificance levels are also returned. A call to G13DMF is usually made prior to calling this routine in order to calculate the sample cross-correlation matrices.

2 Specification

```
 SUBROUTINE G13DNF(K, N, M, IK, R0, R, MAXLAG, PARLAG, X, PVALUE, WORK,
1                  LWORK, IFAIL)
 INTEGER           K, N, M, IK, MAXLAG, LWORK, IFAIL
 real              R0(IK,K), R(IK,IK,M), PARLAG(IK,IK,M), X(M),
1                  PVALUE(M), WORK(LWORK)
```

3 Description

Let $W_t = (w_{1t}, w_{2t}, \dots, w_{kt})^T$, for $t = 1, 2, \dots, n$, denote n observations of a vector of k time series. The partial lag correlation matrix at lag l, $P(l)$, is defined to be the correlation matrix between W_t and W_{t+l}, after removing the linear dependence on each of the intervening vectors $W_{t+1}, W_{t+2}, \dots, W_{t+l-1}$. It is the correlation matrix between the residual vectors resulting from the regression of W_{t+l} on the carriers $W_{t+l-1}, \dots, W_{t+1}$ and the regression of W_t on the same set of carriers; see Heyse and Wei (1985).

$P(l)$ has the following properties.

(i) If W_t follows a vector autoregressive model of order p, then $P(l) = 0$ for $l > p$;

(ii) When $k = 1$, $P(l)$ reduces to the univariate partial autocorrelation at lag l;

(iii) Each element of $P(l)$ is a properly normalized correlation coefficient;

(iv) When $l = 1$, $P(l)$ is equal to the cross-correlation matrix at lag 1 (a natural property which also holds for the univariate partial autocorrelation function).

Sample estimates of the partial lag correlation matrices may be obtained using the recursive algorithm described in Wei (1990). They are calculated up to lag m, which is usually taken to be at most $n/4$. Only the sample cross-correlation matrices ($\hat{R}(l)$, $l = 0, 1, \dots, m$) and the standard deviations of the series are required as input to G13DNF. These may be computed by G13DMF. Under the hypothesis that W_t follows an autoregressive model of order $s-1$, the elements of the sample partial lag matrix $\hat{P}(s)$, denoted by $\hat{P}_{ij}(s)$, are asymptotically Normally distributed with mean zero and variance $1/n$. In addition the statistic

$$X(s) = n \sum_{i=1}^{k} \sum_{j=1}^{k} \hat{P}_{ij}(s)^2$$

has an asymptotic χ^2-distribution with k^2 degrees of freedom. These quantities, X(l), are useful as a diagnostic aid for determining whether the series follows an autoregressive model and, if so, of what order.

4 References

Heyse J F and Wei W W S (1985) The partial lag autocorrelation function *Technical Report No. 32* Department of Statistics, Temple University, Philadelphia

Wei W W S (1990) *Time Series Analysis: Univariate and Multivariate Methods* Addison-Wesley

5 Parameters

1: K – INTEGER *Input*

On entry: the dimension, k, of the multivariate time series.

Constraint: $K \geq 1$.

2: N – INTEGER *Input*

On entry: the number of observations in each series, n.

Constraint: $N \geq 2$.

3: M – INTEGER *Input*

On entry: the number, m, of partial lag correlation matrices to be computed. Note this also specifies the number of sample cross-correlation matrices that must be contained in the array R.

Constraint: $1 \leq M < N$.

4: IK – INTEGER *Input*

On entry: the first dimension of the array R0 and the first and second dimension of the arrays R and PARLAG as declared in the (sub)program from which G13DNF is called.

Constraint: $IK \geq K$.

5: R0(IK,K) – ***real*** array *Input*

On entry: if $i \neq j$, then $R0(i,j)$ must contain the (i,j)th element of the sample cross-correlation matrix at lag zero, $\hat{R}_{ij}(0)$. If $i = j$, then $R0(i,i)$ must contain the standard deviation of the ith series.

6: R(IK,IK,M) – ***real*** array *Input*

On entry: $R(i,j,l)$ must contain the (i,j)th element of the sample cross-correlation at lag l, $\hat{R}_{ij}(l)$, for $l = 1,2,\ldots,m$; $i = 1,2,\ldots,k$; $j = 1,2,\ldots,k$, where series j leads series i (see Section 8).

7: MAXLAG – INTEGER *Output*

On exit: the maximum lag up to which partial lag correlation matrices (along with χ^2-statistics and their significance levels) have been successfully computed. On a successful exit MAXLAG will equal M. If IFAIL = 2 on exit, then MAXLAG will be less than M.

8: PARLAG(IK,IK,M) – ***real*** array *Output*

On exit: $PARLAG(i,j,l)$ contains the (i,j)th element of the sample partial lag correlation matrix at lag l, $\hat{P}_{ij}(l)$, for $l = 1,2,\ldots,MAXLAG$; $i = 1,2,\ldots,k$; $j = 1,2,\ldots,k$.

9: X(M) – ***real*** array *Output*

On exit: $X(l)$ contains the χ^2-statistic at lag l, for $l = 1,2,\ldots,MAXLAG$.

10: PVALUE(M) – ***real*** array *Output*

On exit: $PVALUE(l)$ contains the significance level of the corresponding χ^2-statistic in X for $l = 1,2,\ldots,MAXLAG$.

11: WORK(LWORK) – ***real*** array *Workspace*
12: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which G13DNF is called.

Constraint: $LWORK \geq (5M + 6)K^2 + K$.

13: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, K < 1,
or N < 2,
or M < 1,
or M $\geq$ N,
or IK $<$ K,
or LWORK $< (5\text{M} + 6)\text{K}^2 + \text{K}$.

IFAIL $= 2$

The recursive equations used to compute the sample partial lag correlation matrices have broken down at lag MAXLAG $+ 1$. All output quantities in the arrays PARLAG, X and PVALUE up to and including lag MAXLAG will be correct.

7 Accuracy

The accuracy will depend upon the accuracy of the sample cross-correlations.

8 Further Comments

The time taken is roughly proportional to m^2k^3.

If the user has calculated the sample cross-correlation matrices in the arrays R0 and R, without calling G13DMF, then care must be taken to ensure they are supplied as described in Section 5. In particular, for $l \geq 1$, $\hat{R}_{ij}(l)$ must contain the sample cross-correlation coefficient between $w_{i(t-l)}$ and w_{jt}.

The routine G13DBF computes squared partial autocorrelations for a specified number of lags. It may also be used to estimate a sequence of partial autoregression matrices at lags $1, 2, \ldots$ by making repeated calls to the routine with the parameter NK set to $1, 2, \ldots$. The (i, j)th element of the sample partial autoregression matrix at lag l is given by $W(i, j, l)$ when NK is set equal to l on entry to G13DBF. Note that this is the 'Yule–Walker' estimate. Unlike the partial lag correlation matrices computed by G13DNF, when W_t follows an autoregressive model of order $s - 1$, the elements of the sample partial autoregressive matrix at lag s do not have variance $1/n$, making it very difficult to spot a possible cut-off point. The differences between these matrices are discussed further by Wei (1990).

Note that G13DBF takes the sample cross-covariance matrices as input whereas this routine requires the sample cross-correlation matrices to be input.

9 Example

This program computes the sample partial lag correlation matrices of two time series of length 48, up to lag 10. The matrices, their χ^2-statistics and significance levels and a plot of symbols indicating which elements of the sample partial lag correlation matrices are significant are printed. Three * represent significance at the 0.5% level, two * represent significance at the 1% level and a single * represents significance at the 5% level. The * are plotted above or below the central line depending on whether the elements are significant in a positive or negative direction.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G13DNF Example Program Text
*     Mark 15 Release. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          KMAX, IK, NMAX, MMAX, LWORK
      PARAMETER        (KMAX=3,IK=KMAX,NMAX=100,MMAX=20,LWORK=(5*MMAX+6)
     +                 *KMAX*KMAX+KMAX)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, J, K, M, MAXLAG, N
*     .. Local Arrays ..
      real             PARLAG(IK,IK,MMAX), PVALUE(MMAX), R(IK,IK,MMAX),
     +                 R0(IK,KMAX), W(IK,NMAX), WMEAN(KMAX),
     +                 WORK(LWORK), X(MMAX)
*     .. External Subroutines ..
      EXTERNAL         G13DMF, G13DNF, ZPRINT
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G13DNF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) K, N, M
      IF (K.GT.0 .AND. K.LE.KMAX .AND. N.GE.1 .AND. N.LE.NMAX .AND.
     +    M.GE.1 .AND. M.LE.MMAX) THEN
         DO 20 I = 1, K
            READ (NIN,*) (W(I,J),J=1,N)
   20    CONTINUE
         IFAIL = 0
*
         CALL G13DMF('R-correlation',K,N,M,W,IK,WMEAN,R0,R,IFAIL)
*
         IFAIL = 0
*
         CALL G13DNF(K,N,M,IK,R0,R,MAXLAG,PARLAG,X,PVALUE,WORK,LWORK,
     +               IFAIL)
*
         CALL ZPRINT(K,N,M,IK,PARLAG,X,PVALUE,NOUT)
      END IF
      STOP
*
      END
*
      SUBROUTINE ZPRINT(K,N,M,IK,PARLAG,X,PVALUE,NOUT)
*     .. Scalar Arguments ..
      INTEGER           IK, K, M, N, NOUT
*     .. Array Arguments ..
      real              PARLAG(IK,IK,M), PVALUE(M), X(M)
*     .. Local Scalars ..
      real              C1, C2, C3, C5, C6, C7, CONST, SUM
      INTEGER           I, I2, IFAIL2, J, L, LL
*     .. Local Arrays ..
      CHARACTER*1       CLABS(1), RLABS(1)
      CHARACTER*80      REC(7)
*     .. External Subroutines ..
      EXTERNAL          X04CBF
```

```
*     .. Intrinsic Functions ..
      INTRINSIC          real, SQRT
*     .. Executable Statements ..
*
*     Print the partial lag correlation matrices.
*
      CONST = 1.0e0/SQRT(real(N))
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' PARTIAL LAG CORRELATION MATRICES'
      WRITE (NOUT,*) ' --------------------------------'
      DO 20 L = 1, M
         WRITE (NOUT,99999) ' Lag = ', L
         IFAIL2 = 0
         CALL X04CBF('G','N',K,K,PARLAG(1,1,L),IK,'F9.3',' ','N',RLABS,
     +               'N',CLABS,80,5,IFAIL2)
   20 CONTINUE
      WRITE (NOUT,99998) ' Standard error = 1 / SQRT(N) = ', CONST
*
*     Print indicator symbols to indicate significant elements.
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' TABLES OF INDICATOR SYMBOLS'
      WRITE (NOUT,*) ' ---------------------------'
      WRITE (NOUT,99999) ' For Lags 1 to ', M
*
*     Set up annotation for the plots.
*
      WRITE (REC(1),99997) '              0.005  :'
      WRITE (REC(2),99997) '        +     0.01   :'
      WRITE (REC(3),99997) '              0.05   :'
      WRITE (REC(4)(1:23),99997) '   Sig. Level        :'
      WRITE (REC(4)(24:),99997) '- - - - - - - - - -  Lags'
      WRITE (REC(5),99997) '              0.05   :'
      WRITE (REC(6),99997) '        -     0.01   :'
      WRITE (REC(7),99997) '              0.005  :'
*
*     Set up the critical values
*
      C1 = 3.29e0*CONST
      C2 = 2.58e0*CONST
      C3 = 1.96e0*CONST
      C5 = -C3
      C6 = -C2
      C7 = -C1
*
      DO 120 I = 1, K
         DO 100 J = 1, K
            WRITE (NOUT,*)
            IF (I.EQ.J) THEN
               WRITE (NOUT,99996) ' Auto-correlation function for',
     +           ' series ', I
            ELSE
               WRITE (NOUT,99995) ' Cross-correlation function for',
     +           ' series ', I, ' and series', J
            END IF
            DO 60 L = 1, M
               LL = 23 + 2*L
               SUM = PARLAG(I,J,L)
*
*              Clear the last plot with blanks
*
               DO 40 I2 = 1, 7
                  IF (I2.NE.4) REC(I2) (LL:LL) = ' '
   40          CONTINUE
*
*              Check for significance
*
               IF (SUM.GT.C1) REC(1) (LL:LL) = '*'
               IF (SUM.GT.C2) REC(2) (LL:LL) = '*'
               IF (SUM.GT.C3) REC(3) (LL:LL) = '*'
               IF (SUM.LT.C5) REC(5) (LL:LL) = '*'
```

```
                IF (SUM.LT.C6) REC(6) (LL:LL) = '*'
                IF (SUM.LT.C7) REC(7) (LL:LL) = '*'
   60       CONTINUE
*
*           Print
*
            DO 80 I2 = 1, 7
               WRITE (NOUT,99997) REC(I2)
   80       CONTINUE
  100    CONTINUE
  120 CONTINUE
*
*     Print the chi-square statistics and p-values.
*
      WRITE (NOUT,*)
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' Lag     Chi-square statistic     P-value'
      WRITE (NOUT,*) ' ---     --------------------     -------'
      WRITE (NOUT,*)
      DO 140 L = 1, M
         WRITE (NOUT,99994) L, X(L), PVALUE(L)
  140 CONTINUE
      RETURN
*
99999 FORMAT (/1X,A,I2)
99998 FORMAT (/1X,A,F5.3,A)
99997 FORMAT (1X,A)
99996 FORMAT (//1X,A,A,I2,/)
99995 FORMAT (//1X,A,A,I2,A,I2,/)
99994 FORMAT (1X,I4,10X,F8.3,11X,F8.4)
      END
```

9.2 Program Data

```
G13DNF Example Program Data
2 48 10 :  K, no. of series,  N, no. of obs in each series,  M, no. of lags
-1.490 -1.620  5.200  6.230  6.210  5.860  4.090  3.180
 2.620  1.490  1.170  0.850 -0.350  0.240  2.440  2.580
 2.040  0.400  2.260  3.340  5.090  5.000  4.780  4.110
 3.450  1.650  1.290  4.090  6.320  7.500  3.890  1.580
 5.210  5.250  4.930  7.380  5.870  5.810  9.680  9.070
 7.290  7.840  7.550  7.320  7.970  7.760  7.000  8.350
 7.340  6.350  6.960  8.540  6.620  4.970  4.550  4.810
 4.750  4.760 10.880 10.010 11.620 10.360  6.400  6.240
 7.930  4.040  3.730  5.600  5.350  6.810  8.270  7.680
 6.650  6.080 10.250  9.140 17.750 13.300  9.630  6.800
 4.080  5.060  4.940  6.650  7.940 10.760 11.890  5.850
 9.010  7.500 10.020 10.380  8.150  8.370 10.730 12.145 : End of time series
```

9.3 Program Results

```
 G13DNF Example Program Results

  PARTIAL LAG CORRELATION MATRICES
  --------------------------------

  Lag =  1
           0.736    0.174
           0.211    0.555

  Lag =  2
          -0.187   -0.083
          -0.180   -0.072

  Lag =  3
           0.278   -0.007
           0.084   -0.213

  Lag =  4
          -0.084    0.227
```

```
         0.128   -0.176

Lag =  5
         0.236    0.238
        -0.047   -0.046

Lag =  6
        -0.016    0.087
         0.100   -0.081

Lag =  7
        -0.036    0.261
         0.126    0.012

Lag =  8
         0.077    0.381
         0.027   -0.149

Lag =  9
        -0.065   -0.387
         0.189    0.057

Lag = 10
        -0.026   -0.286
         0.028   -0.173

Standard error = 1 / SQRT(N) = 0.144

TABLES OF INDICATOR SYMBOLS
---------------------------

For Lags 1 to 10


Auto-correlation function for series  1

              0.005  : *
        +     0.01   : *
              0.05   : *
   Sig. Level        : - - - - - - - - - -  Lags
              0.05   :
        -     0.01   :
              0.005  :


Cross-correlation function for series  1 and series 2

              0.005  :
        +     0.01   :              *
              0.05   :              *
   Sig. Level        : - - - - - - - - - -  Lags
              0.05   :                * *
        -     0.01   :                *
              0.005  :


Cross-correlation function for series  2 and series 1

              0.005  :
        +     0.01   :
              0.05   :
   Sig. Level        : - - - - - - - - - -  Lags
              0.05   :
        -     0.01   :
              0.005  :
```

```
 Auto-correlation function for series  2

              0.005  : *
        +     0.01   : *
              0.05   : *
    Sig. Level       : - - - - - - - - - - -  Lags
              0.05   :
        -     0.01   :
              0.005  :


 Lag     Chi-square statistic     P-value
 ---     --------------------     -------

   1           44.362              0.0000
   2            3.824              0.4304
   3            6.219              0.1834
   4            5.094              0.2778
   5            5.609              0.2303
   6            1.170              0.8830
   7            4.098              0.3929
   8            8.371              0.0789
   9            9.244              0.0553
  10            5.435              0.2455
```